

HEWLETT-PACKARD

HP-71 to HP 3000 and HP 1000 File Transfers

Summary

This application note provides an example of how to transfer text files between the HP-71 Handheld Computer and the HP 3000 and HP 1000 computers, and also provides dumb terminal capabilities to nearly any host computer. Files can be transferred either from the HP-71 to the host, or from the host to the HP-71. The procedure described does not require any special program on the host, but instead relies on the host's editor to perform the transfers.

The file transfer capability described in this note makes the HP-71 a simple solution to remote data capture applications involving an HP 3000 or HP 1000 host. The HP 82164A HP-IL/RS-232C Interface is used to connect to a terminal port on the host computer.

A program listing is included which will provide the following functions:

- Automatic assignment of the HP-IL.
- Interactive configuration of the HP 82164A HP-IL/RS-232C converter.
- Use of a data file to "remember" the communications protocol.
- "Dumb" terminal emulation.
- Text file upload to a host HP 3000 or HP 1000 computer.
- Text file download from a host HP 3000 or HP 1000 computer.
- Optional printer logging of all communications.

Equipment

Required:

HP-71 Handheld Computer
HP 82164A HP-IL/RS-232C Interface
HP-IL cables
HP 82401A HP-IL Interface

Optional:

HP-IL Video Interface with monitor
HP-IL Printer


Operation

1. Connect all HP-IL peripherals and turn them on. Load the BASIC program listed in this note into an HP-71 file called 'TERMHP71'. Type: run TERMHP71 **[ENDLINE]**

After the program has assigned the HP-IL, it will clear the display, and then display "HP71 file transfer".

2. If the program has not been executed previously, it will ask you to "Set Configuration." The communications protocol configuration is stored in a BASIC data file called "CONFGCOM". If this file does not exist, or does not contain proper data, the program will require you to specify the following parameters:

- Host Computer. This tells the program which host (either HP 3000 or HP 1000) it will be communicating with. The program will display "Host: 1000 or 3000". Pressing either 1 or 3 will select the HP 1000 or the HP 3000 respectively.
- Baud Rate. This is the data communications speed (in bits per second). The HP-71 will display "Select Baud Rate" and then display "Baud: 300 1200 2400 4800 9600". Pressing the first digit associated with any of the five choices will select that rate.
- Parity. This specifies the type of error detection used, if any. The HP-71 will first display "Select Parity", then "Parity: Even Odd 0



1 None". Pressing the upper-case letter associated with one of the five choices will select that parity.

- Software Protocol. This selects the type of software handshake the host computer expects to see from the terminal port being used. The HP-71 will display "Select Protocol", then "Xon/Xoff Enq/Ack Both None All". Pressing the upper-case letter associated with one of the five choices will select that protocol. Selecting the "Xon/Xoff" protocol is the same as pressing **[CTL] [S]** and **[CTL] [Q]** to stop and start data transmission from the host. The "Enq/Ack" is a transmitter protocol with no prompt character observed. The "Both" option selects both Xon/Xoff and Enq/Ack with no prompt character, while the "All" option specifies Xon/Xoff and Enq/Ack with a prompt character. See the "Protocol Explanation" section of this note for a complete description of each of these protocols.

To communicate with an HP 3000, select the "All" option. To communicate with an HP 1000, select the "Both" option. To communicate with other hosts, select from the other options which are available.

All of these parameters can vary between different host computers, and even between different ports of the same host. As a general guide, select the highest baud rate the host port will support.

If you are unsure of the required parity, select the "None" option. For the files to upload properly, it is essential that the proper software protocol be selected.

Pressing **[F1]** while in terminal mode will return you to these configuration menus, allowing you to change the parameters "on the fly."

3. After the configuration has been set, the HP-71 will display "Terminal 1000 Ready . . ." or "Terminal 3000 Ready . . ." At this point, the HP-71 is acting as a dumb terminal. Any characters pressed on the keyboard will be transmitted, and any characters received from the host will be displayed. Pressing **[ENDLINE]** usually gets the attention of the host, and causes it to send you a request to log on. If you experience problems either in receiving the log-on message or in logging on, carefully review the communications protocol you have specified.

4. After you have logged on, you can initiate the file upload operation by pressing **[F2]**. The program will then prompt you with: "HP-71 Source File:". Key in the name of the HP-71 text file you wish to send to the host, and press **[ENDLINE]**. The program will then prompt you with: "3000 Destination Name:" or "1000 Destination Name:". Key in the name you want the file stored under on the host. This file MUST NOT already exist on the host. The program will display the messages it receives from the host as it calls the editor, sends the file, keeps the file, and exits the editor.

5. The file download operation can be initiated by pressing **[F3]**. The program will prompt you for:

"HP-71 Destination File:". Key in the name of the HP-71 text file you wish to store the data into, and press **[ENDLINE]**. The program will then prompt you for either "1000 Source Name?" or "3000 Source Name?". Key in the name of the file on the host which you want to download and press **[ENDLINE]**. The program will display all the messages it receives from the host as it runs the editor.

6. When the program is waiting for a response from the host in either the upload or download routines, you can abort the transfer operation and return to the terminal level by pressing **[F4]** on the HP-71. This is a way out if you specify a bad host file name, or receive some other error message from the host.

Selected Commands

In addition to the **[F1]** through **[F4]** keystrokes, the following command keys are also defined:

[F5]

Toggles the printer either on or off.

[F6]

Toggles between the LCD on the HP-71 and the video interface on HP-IL.

[F7]

Sends a BREAK to the host.

All other keys on the alphanumeric keypad are transmitted to the host.

Protocol Explanation

This section provides a brief explanation of how each of the different protocol options function, and an explanation of how the HP 3000 Editor, and EDIT 1000 expect the terminal to behave.

The XON/XOFF protocol is called a "receiver" protocol. The computer receiving data can halt the data transmission by transmitting an XOFF character, and can resume transmission by sending an XON

character to the computer which is sending data. Many people have used this handshake manually from the keyboard without realizing that it is the same as Xon/Xoff.

The 'ENQ/ACK' protocol is called a "transmitter" protocol. The computer which is transmitting data must initiate the handshake by sending an ENQ (enquire) character after each "block" of data. The computer which is receiving data will respond with an ACK (acknowledge) character when it is ready for more data.

The 'All' option enables the HP 82164A HP-IL/RS-232C Interface to observe the Xon/Xoff handshake, Enq/Ack handshake, and a prompt character. After receiving a line of data, the HP 3000 will send a prompt character to the terminal when it is ready for more data. The HP 3000 actually sends two prompt characters, usually either a colon(:) or a slash(/) followed by an XON character. It is this XON character for which the HP-IL/RS-232C Interface waits before sending the next line. If this option is not enabled when doing a file upload, the HP-IL/RS-232C Interface will send the next line before the HP 3000 is ready, and data will almost certainly be lost. The "All" option is the only mode in which the HP-IL/RS-232C Interface is enabled to observe a prompt character handshake.

The 'Both' option enables both Xon/Xoff and Enq/Ack, but not a prompt character, while the "None" option tells the HP-IL/RS-232C Interface to ignore all handshakes, and not to send any handshake characters. The HP 1000 uses both Xon/Xoff and Enq/Ack to control the transmission of data, so the "Both" option works best.

Note: The following program is available on magnetic card through the Users' Library at a cost of \$15.00 each. Order 71-03000 from:

Hewlett-Packard
Users' Library, Dept. 39UL
1000 N.E. Circle Blvd.
Corvallis, OR 97330

Programming

Line #	Keystrokes	Comment
10	DESTROY ALL	
20	SFLAG -23	Terminate Enters on EOT.
30	DIM A\$(256),B\$(256),K2\$(40)	
40	P3=1	
50	A=DEVADDR("RS232")	Find the RS-232C interface.
60	IF A=-1 THEN BEEP (@ DISP 'No RS232 interface' (@ END	Error not found.
70	F=DEVADDR("DISPLAY")	Find the display device.
80	IF F=-1 THEN DELAY 0,0 ELSE DELAY 0,INF	Set delay appropriately.
90	PRINTER IS *	Print to display only.
100	PWIDTH 80	Set the printer/display width.
110	DISP CHR\$(27)&"E"	Clear the display.
120	DISP 'HP71B file transfer'	Welcome message.
130	ON ERROR GOTO 150	
140	CREATE DATA CONFGCOM,5,20	Create the configuration file.
150	OFF ERROR	
160	ASSIGN #1 TO CONFGCOM	
170	ON ERROR GOTO 180 (@ GOTO 190	
180	OFF ERROR (@ BEEP (@ DISP "Set Configuration" (@ WAIT .3 (@ GOTO 'CONFIG'	Branch to configure for proper host.
190	READ #1,1;H\$	Read name of host: either HP 1000 or HP 3000.
200	IF H\$="1000" THEN H=1 ELSE H=0	Set the host 'type' flag.
210	READ #1,2;S\$	Read the baud rate specifier.
220	READ #1,3;B\$	Read the parity select specifier.
230	READ #1,4;C\$	Read the software protocol specifier.
240	OFF ERROR (@ GOTO 530	Go and write the control registers.
250	'CONFIG':	This is the "configure" section.
260	DISP 'Host:1000 or 3000?';	Select the host.
270	ON POS("13",KEY\$)+1 GOTO 270,280,290	
280	H\$="1000" (@ H=1 (@ GOTO 300	
290	H\$="3000" (@ H=0	
300	DISP (@ DISP 'Host Selected: ';H\$	
310	PRINT #1,1;H\$	Store the host type.
320	DISP (@ DISP (@ DISP "Select Baud Rate"	Select the baud rate.
330	DISP "300 1200 2400 4800 9600";	
340	ON POS("31249",KEY\$)+1 GOSUB 1640,1650,1660,1670,1680,1690	
350	DISP (@ DISP "Selected Baud Rate: ";S1\$	
360	PRINT #1,2;S\$	Store the baud rate.
370	B1\$="" (@ DISP (@ DISP (@ DISP 'Select Parity'	Select the parity.
380	DISP 'Parity: Even Odd 0 1 None';	
390	ON POS("EO01N",UPRC\$(KEY\$))+1 GOSUB 1700,1710,1720,1730,1740,1750	
400	DISP (@ DISP 'Parity Selected: ';B1\$	
410	PRINT #1,3;B\$	Store the parity.

Line #	Keystrokes	Comment
420	DISP @ DISP 'Select Protocol'	Set the software protocol.
430	DISP 'Xon/xoff Enq/ack Both None All';	The 'All' option includes prompt character.
440	ON POS('NXEBA',UPRC\$(KEY\$))+1 GOTO 450,460,470,480,490,500	
450	GOTO 440	
460	C\$='C0;' @ C1\$='No Protocol' @ GOTO 510	
470	C\$='C0;C2;' @ C1\$='Xon/Xoff' @ GOTO 510	
480	C\$='C0;C1;' @ C1\$='Enq/Ack' @ GOTO 510	
490	C\$='C0;C1;C2;' @ C1\$='Both Xon/Xoff-Enq/Ack' @ GOTO 510	
500	C\$='C0;C1;C2;C4;' @ C1\$='Xon/off-Enq/Ack-Prompt'	
510	DISP @ DISP 'Protocol: ';C1\$	
520	PRINT #1,4;C\$	Store the software protocol.
		Set the configuration in the RS-232C converter. Disable service request; set baud, parity, protocol, clear buffer.
530	G\$='SS0;SW1;LI1;LI3;R0;R1;'	Sets 7 data bits, one stop bit, DSR true, RTS true.
540	REMOTE	Put the RS-232C into remote mode for setting configuration.
550	OUTPUT :A USING 'K';'SE0;'&G\$\$&B\$&C\$&'R';	Send the configuration information.
560	LOCAL	Send a not remote enable (nre) command to disable remote mode.
570	ASSIGN #1 TO *	Close CONFGCOM file.
580	K2\$=" f1 f2 f3 f4 f5 f6 f7 #38 #103"	Create string of recognized special characters. f1 = Change CONFGCOM file. f2 = Upload to host. f3 = Download from host. f4 = Return to terminal. f5 = Toggle printer. f6 = Toggle video. f7 = Send break to host. #38 = Endline key sends CHR\$(13). #103 = Send backspace CHR\$(8).
590	DISP @ DISP 'Terminal ';H\$; Ready...'	Ready to talk to host.
600	S = 1	
610	K\$=KEY\$ @ IF K\$#" THEN GOSUB 770	If key is pressed then decode the key.
620	ENTER :A ;A\$	Get data from host.
630	IF F THEN GOTO 670	
640	IF A\$=CHR\$(13) THEN PRINT @ GOTO 610	

Line #	Keystrokes	Comment
650	PRINT A\$;	
660	GOTO 610	
670	FOR I=1 TO LEN(A\$)	
680	IF A\$[I,I]# ' ' THEN 710	Check for space.
690	IF NOT S THEN PRINT ' '; (@ S=1	
700	GOTO 750	
710	IF A\$[I,I]=CHR\$(10) THEN PRINT (@ GOTO 750	
720	IF A\$[I,I]=CHR\$(7) THEN BEEP (@ GOTO 750	
730	IF A\$[I,I]<=CHR\$(31) THEN 750	
740	PRINT A\$[I,I]; (@ S=0	
750	NEXT I	
760	GOTO 610	
		Routine to process keys hit while in terminal mode.
770	IF LEN(K\$)>1 THEN "KEYCODE"	Process special keystroke.
780	OUTPUT :A USING '#,K';K\$; (@ RETURN	Send the key to host.
		Download a file from the host.
790	ON ERROR GOTO 800 (@ GOTO 810	Branch for bad file.
800	BEEP (@ DISP N\$;' Is A Bad File Name'	Display message.
810	INPUT '71 Destination Name ?';N\$	Prompt for file name.
820	ASSIGN #3 TO N\$	
830	OFF ERROR	
840	IF NOT H THEN 920	If host is HP 3000 then skip HP 1000 portion.
		Run the HP 1000 editor and tell the HP 1000 to list the file.
850	INPUT '1000 Source Name ?';N1\$	
860	GOSUB 1880	Gosub to run the HP 1000 editor.
870	OUTPUT :A USING '#,K';'fi,'&N1\$&CHR\$(13);	Get the file.
880	GOSUB 1770	Wait for the editor prompt.
890	OUTPUT :A USING '#,K';'1\$L'&CHR\$(13);	List all lines in the file.
900	ENTER :A ;A\$	Read back the echo ('1\$L').
910	GOTO 980	Go read and store all lines in the file.
		Run the HP 3000 editor.
920	INPUT '3000 Source Name ?';N1\$	
930	GOSUB 1950	Run the editor.
940	OUTPUT :A USING 'K';'t '&N1\$	Make the named file the working text file.
950	W\$="/" (@ GOSUB 1770	Wait for the HP 3000 to get the file.
960	OUTPUT :A USING 'K';'list all, unnumbered'	Tell it to send the whole file.
970	ENTER :A ;A\$	Read back the echo ('list all').
980	DISP (@ DISP 'Receiving Host file';	Display the message.
990	B\$=""	Null string for beginning line.
1000	ENTER :A ;A\$	Enter the string.
1010	B\$=B\$&A\$	Append to main string.
1020	IF KEY\$="f4" THEN PRINT 'Transfer Aborted' (@ RETURN	Check for abort key.

Line #	Keystrokes	Comment
1030	IF NOT LEN(B\$) THEN 1000	Loop if no length.
1040	IF H AND POS(B\$,CHR\$(13)&"") THEN PRINT #3;B\$[1,LEN(B\$)-3] @ GOTO 1080	If HP 1000, then check for editor prompt.
1050	IF POS(B\$,CHR\$(10)&"") THEN PRINT #3;B\$[1,LEN(B\$)-3] @ GOTO 1080	Check for HP 3000 editor prompt.
1060	R5 = POS(B\$,CHR\$(10)) @ IF NOT R5 THEN 1000	Check for end of line.
1070	PRINT #3;B\$[1,R5-2] @ B\$ = B\$[R5 + 1] @ GOTO 1050	Write line to file.
1080	IF H THEN OUTPUT :A USING 'K';'A' @ GOTO 1100	Exit the HP 1000 editor.
1090	OUTPUT :A USING 'K';'exit'	Exit the HP 3000 editor.
1100	ASSIGN #3 TO *	Close the HP-71 file.
1110	DISP @ DISP 'File Download Complete' @ DISP	Display message.
1120	RETURN	Back to terminal.
1130	IF P5 THEN PRINTER IS * @ P5 = 0 ELSE PRINTER IS :PRINTER @ P5 = 1	Printer toggle here.
1140	RETURN	
1150	IF D5 THEN DISPLAY IS * @ D5 = 0 @ DELAY 0,0 ELSE DISPLAY IS :DISPLAY @ D5 = 1 @ DELAY 0,INF	Display toggle here.
1160	RETURN	
1170	DISP @ INPUT "HP-71 Source File: ";N\$	Upload a file to the host. Input file name.
1180	ON ERROR GOTO 1570	Error if file not good.
1190	ASSIGN #3 TO N\$	
1200	OFF ERROR	
1210	IF H THEN 1400	If host is HP 1000 then skip the HP 3000 section.
1220	DISP @ INPUT '3000 Destination Name: ';N1\$	This is the HP 3000 section.
1230	GOSUB 1950	
1240	OUTPUT :A USING 'K';'aq'	Invoke add-quiet mode.
1250	ON ERROR GOTO 1310	Error if end of file.
1260	DISP @ DISP 'Sending File ';N\$;	Display message.
1270	READ #3;A\$	Read the file.
1280	OUTPUT :A USING 'K';A\$	Output to host.
1290	ENTER :A USING "#,K";A\$	Enter echoed characters.
1300	GOTO 1270	
1310	OFF ERROR	
1320	ASSIGN #1 TO *	Close the file.
1330	OUTPUT :A USING 'K';CHR\$(25)	<Ctrl 'Y'> to exit the add-quiet mode.
1340	W\$ = ' ' @ GOSUB 1770	Wait for editor prompt.
1350	OUTPUT :A USING 'K';'keep '&N1\$&',unnumbered'	Store text in text editor file.
1360	GOSUB 1770	
1370	OUTPUT :A USING 'K';'exit'	Exit the editor.

Line #	Keystrokes	Comment
1380	DISP 'Upload Complete' @ DISP	Display message.
1390	RETURN	Return to terminal.
1400	DISP @ INPUT '1000 Destination Name: ';N1\$	HP 1000 upload.
1410	GOSUB 1880	Run the HP 1000 editor.
1420	ON ERROR GOTO 1510	
1430	W\$=CHR\$(13)&''	
1440	DISP @ DISP 'Sending File ';N\$	
1450	P3=0	
1460	READ #3;A\$	Read HP-71 file.
1470	IF A\$[1,1]# ' ' THEN A\$=' '&A\$	Force first character to be a space.
1480	OUTPUT :A USING '#,K';A\$&CHR\$(13);	Send it to the HP 1000.
1490	GOSUB 1760	Wait for prompt.
1500	GOTO 1460	
1510	OFF ERROR	
1520	ASSIGN #3 TO *	Close the file.
1530	P3=1	
1540	OUTPUT :A USING '#,K';'ec,'&N1\$&CHR\$(13);	Create the file and exit editor.
1550	DISP 'Upload Complete' @ DISP	Display message.
1560	RETURN	Return to terminal level.
1570	BEEP @ DISP 'File ';N\$;' not text'	
1580	OFF ERROR	
1590	RETURN	
1600	SEND UNL UNT LISTEN A DDL 3	Set break.
1610	WAIT 2	Hold it !!
1620	SEND UNL UNT LISTEN A DDL 4	Release break.
1630	BEEP @ RETURN	Beep to signal end of break.
		Subroutines to configure system.
1640	POP @ GOTO 340	Baud rate.
1650	S\$='SB6;' @ S1\$='300' @ RETURN	
1660	S\$='SB8;' @ S1\$='1200' @ RETURN	
1670	S\$='SBA;' @ S1\$='2400' @ RETURN	
1680	S\$='SBC;' @ S1\$='4800' @ RETURN	
1690	S\$='SBE;' @ S1\$='9600' @ RETURN	
1700	POP @ GOTO 390	Select parity.
1710	B\$='P0;SP1;' @ B1\$='Even' @ RETURN	
1720	B\$='P1;SP1;' @ B1\$='Odd' @ RETURN	
1730	B\$='P2;SP1;' @ B1\$='Always 0' @ RETURN	
1740	B\$='P3;SP1;' @ B1\$='Always 1' @ RETURN	
1750	B\$='P4;SW0;' @ B1\$='None' @ RETURN	
1760	P2=1 @ GOTO 1780	Routine to wait for a specified character.
1770	P2=0	
1780	ENTER :A ;A\$	
1790	IF NOT LEN(A\$) THEN 1830	
1800	PRINT A\$;	

Line #	Keystrokes	Comment
1810	IF H = 1 AND POS(A\$,W\$) THEN RETURN	
1820	IF A\$(LEN(A\$))=W\$ THEN RETURN	
1830	IF KEY\$#"f4" THEN 1780	If transfer is to be aborted.
1840	PRINT @ PRINT "Transfer Aborted"	
1850	IF P2 THEN POP	Do an extra pop if two levels deep.
1860	POP	
1870	RETURN	
1880	OUTPUT :A USING 'K';'ru,edit'	Subroutine to run HP 1000 editor.
1890	W\$=CHR\$(27)&'@' @ GOSUB 1760	Wait for 26xx query.
1900	OUTPUT :A USING "#,K" ;CHR\$(13);	Send a carriage return.
1910	GOSUB 1760	
1920	OUTPUT :A USING "#,K" ;CHR\$(13);	Send a carriage return.
1930	W\$=CHR\$(13)&'@' @ GOSUB 1760	Wait for editor prompt.
1940	RETURN	
1950	OUTPUT :A USING 'K';'run editor.pub.sys'	Subroutine to run HP 3000 editor.
1960	W\$='@' @ GOSUB 1760	
1970	RETURN	
1980	'KEYCODE':	Special key service routines.
1990	ON POS(K2\$,K\$)/4+1 GOTO 2000, 2010,2020,2030,2040,2050,2060, 2070,2080,2090	
2000	K\$="" @ RETURN	
2010	K\$="" @ POP @ ASSIGN #1 TO CONFGCOM @ GOTO 260	
2020	K\$="" @ GOTO 1170	
2030	K\$="" @ GOTO 810	
2040	K\$="" @ RETURN	
2050	K\$="" @ GOTO 1130	
2060	K\$="" @ GOTO 1150	
2070	K\$="" @ GOTO 1600	
2080	K\$=CHR\$(13) @ GOTO 780	
2090	K\$=CHR\$(8) @ PRINT '<'; @ GOTO 780	

Hewlett-Packard
Handheld Products Operation
1000 N.E. Circle Blvd.
Corvallis, Oregon 97330

For additional information, visit your
nearest HP dealer. For the location and
number of the dealer nearest you, call
toll-free 1-800-FOR-HPPC (1-800-
367- 4772).



Scan Copyright ©
The Museum of HP Calculators
www.hpmuseum.org

Original content used with permission.

Thank you for supporting the Museum of HP
Calculators by purchasing this Scan!

Please to not make copies of this scan or
make it available on file sharing services.